

Example 2c: Practice with Interactions among Categorical Predictors in General Linear Models (as estimated using restricted maximum likelihood in MIXED)

The models for this example come from Hoffman (2015) chapter 2. We will be examining the extent to which cognition (as measured by an information test outcome) can be predicted from age (centered at 85 years) grip strength (centered at 9 pounds), sex (with men as the reference group) and dementia status (none, future, current, with none as the reference) in a sample of 550 older adults. In example 2a we examined main effects only; in example 2b we examined interactions involving age, grip strength, and sex (a binary predictor that was treated as continuous). Now we examine the interaction of sex by dementia group when treated as “continuous” via manual contrasts or as “categorical” via CLASS.

First, **renouncing ANOVA dogma:** You can ask for Type I error correction to pairwise follow-ups if desired, although I don’t believe they should have any special status relative to any other default-provided p -values because all group differences and their SEs are already predicted by the model.

In addition, it is possible to have significant pairwise group comparisons even if the “omnibus” test is not significant because it is evaluated per df. So if one group is very different than all the others, then the average effect per df may not be different than 0 (but you’d be missing the real story by only examining the “omnibus” result). So I do not think the omnibus F-test should be the gate-keeper to examining group comparisons. But just remember, for every 20 p -values, one is significant by chance!

SAS syntax for data manipulation:

```
* Import chapter 2 example data into work library;
DATA work.Chapter2; SET filesave.SAS_Chapter2;
* Center continuous predictors;
age85 = age - 85;
grip9 = grip - 9;
* Creating manual contrasts for dementia groups (to be treated as continuous);
  IF demgroup=1 THEN DO; demNF=0; demNC=0; END; * None group;
ELSE IF demgroup=2 THEN DO; demNF=1; demNC=0; END; * Future group;
ELSE IF demgroup=3 THEN DO; demNF=0; demNC=1; END; * Current group;
* Labeling new variables;
LABEL
age85= "age85: Age in Years (0=85)"
grip9= "grip9: Grip Strength in Pounds (0=9)"
sexMW= "sexMW: Sex (0=M, 1=W)"
demNF= "demNF: Dementia Contrast for None=0 vs Future=1"
demNC= "demNC: Dementia Contrast for None=0 vs Current=1";
demgroup= "dementia group NFC";
RUN;

* Creating value labels to use in CLASS statement;
PROC FORMAT;
VALUE FDemGroup 1="1None" 2="2Future" 3="3Current";
VALUE FSex 0="0Men" 1="1Women";
RUN;
```

We will use this table of estimated cell and marginal means to understand what the ANOVA F-tests provided by default tell us (and why they are basically useless given a significant interaction).

Equation 2.13, adding sex*dementia group:

$$\begin{aligned} \text{Cognition}_i = & \beta_0 + \beta_1 (\text{Age}_i - 85) + \beta_2 (\text{Grip}_i - 9) + \beta_3 (\text{SexMW}_i) \\ & + \beta_4 (\text{DemNF}_i) + \beta_5 (\text{DemNC}_i) + \beta_6 (\text{Age}_i - 85)(\text{Grip}_i - 9) \\ & + \beta_7 (\text{SexMW}_i)(\text{DemNF}_i) + \beta_8 (\text{SexMW}_i)(\text{DemNC}_i) + e_i \end{aligned}$$

Dementia Group	Men	Women	Marginal Mean
None	29.07	26.20	27.63
Future	23.01	20.30	21.66
Current	17.10	6.35	11.72
Marginal Mean	23.03	17.62	$\sigma_e^2 = 85.97$

SAS syntax and output: sex and dementia group are both treated as continuous (dummy codes):

```
TITLE1 'Eq 2.13: Adding Sex by Dementia Interaction';
TITLE2 'Continuous Sex (0=Men), Continuous Dementia (0=None)';
PROC MIXED DATA=work.Chapter2 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
MODEL cognition = age85 grip9 age85*grip9 sexMW demNF demNC
              sexMW*demNF sexMW*demNC / SOLUTION CHISQ DDFM=BW;
CONTRAST 'Model R2 Test' age85 1, grip9 1, age85*grip9 1, sexMW 1, demNF 1, demNC 1,
              sexMW*demNF 1, sexMW*demNC 1 / CHISQ;
CONTRAST 'Omnibus Dementia*Sex Interaction' sexMW*demNF 1, sexMW*demNC 1 / CHISQ;

* Cell means (predicted intercepts);
ESTIMATE 'Men None' intercept 1 sexMW 0 demNF 0 demNC 0 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE 'Women None' intercept 1 sexMW 1 demNF 0 demNC 0 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE 'Men Future' intercept 1 sexMW 0 demNF 1 demNC 0 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE 'Women Future' intercept 1 sexMW 1 demNF 1 demNC 0 sexMW*demNF 1 sexMW*demNC 0;
ESTIMATE 'Men Current' intercept 1 sexMW 0 demNF 0 demNC 1 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE 'Women Current' intercept 1 sexMW 1 demNF 0 demNC 1 sexMW*demNF 0 sexMW*demNC 1;

* Simple effects of sex and dementia group;
ESTIMATE 'Sex Diff for None' sexMW 1 demNF 0 demNC 0 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE 'Sex Diff for Future' sexMW 1 demNF 0 demNC 0 sexMW*demNF 1 sexMW*demNC 0;
ESTIMATE 'Sex Diff for Current' sexMW 1 demNF 0 demNC 0 sexMW*demNF 0 sexMW*demNC 1;
ESTIMATE 'None-Fut Diff in Men' sexMW 0 demNF 1 demNC 0 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE 'None-Fut Diff in Women' sexMW 0 demNF 1 demNC 0 sexMW*demNF 1 sexMW*demNC 0;
ESTIMATE 'None-Cur Diff in Men' sexMW 0 demNF 0 demNC 1 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE 'None-Cur Diff in Women' sexMW 0 demNF 0 demNC 1 sexMW*demNF 0 sexMW*demNC 1;
ESTIMATE 'Fut-Current Diff in Men' sexMW 0 demNF -1 demNC 1 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE 'Fut-Current Diff in Women' sexMW 0 demNF -1 demNC 1 sexMW*demNF -1 sexMW*demNC 1;

* Diffs in simple effects = interactions;
ESTIMATE 'A: Sex Effect differ between None and Future?' sexMW*demNF 1 sexMW*demNC 0;
ESTIMATE 'A: None-Future Effect differ by Sex?' sexMW*demNF 1 sexMW*demNC 0;
ESTIMATE 'B: Sex Effect differ between None and Current?' sexMW*demNF 0 sexMW*demNC 1;
ESTIMATE 'B: None-Current Effect differ by Sex?' sexMW*demNF 0 sexMW*demNC 1;
ESTIMATE 'C: Sex Effect differ between Future and Current?' sexMW*demNF -1 sexMW*demNC 1;
ESTIMATE 'C: Future-Current Effect differ by Sex?' sexMW*demNF -1 sexMW*demNC 1;
RUN; TITLE2;
```

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	29.0701	0.7485	541	38.84	<.0001
age85	-0.3348	0.1199	541	-2.79	0.0054
grip9	0.6179	0.1481	541	4.17	<.0001
age85*grip9	0.1222	0.04035	541	3.03	0.0026
sexMW	-2.8756	1.0112	541	-2.84	0.0046
demNF	-6.0559	1.6351	541	-3.70	0.0002
demNC	-11.9707	2.2450	541	-5.33	<.0001
sexMW*demNF	0.1643	2.0705	541	0.08	0.9368
sexMW*demNC	-7.8751	3.0245	541	-2.60	0.0095

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
ge85	1	541	7.80	7.80	0.0052	0.0054
grip9	1	541	17.41	17.41	<.0001	<.0001
age85*grip9	1	541	9.16	9.16	0.0025	0.0026
sexMW	1	541	8.09	8.09	0.0045	0.0046
demNF	1	541	13.72	13.72	0.0002	0.0002
demNC	1	541	28.43	28.43	<.0001	<.0001
sexMW*demNF	1	541	0.01	0.01	0.9368	0.9368
sexMW*demNC	1	541	6.78	6.78	0.0092	0.0095

All tests are df=1 because all fixed effects are single df (treated as separate).

Dementia Group Means	Men Mean	Women Mean	MARGINAL MEAN	Sex Difference
None Mean	29.07	26.20	27.62	-2.87 (p=.0046)
Future Mean	23.01	20.30	21.66	-2.71 (p=.1485)
Current Mean	<u>17.10</u>	<u>6.35</u>	11.72	-10.75 (p=.0002)
MARGINAL	23.03	17.62		-5.45 (p<.0001)

Dementia Group Differences	Men Diff	Women Diff	MARGINAL DIFF	Simple Effect Difference
None-Future Diff	-6.06 (p=.0002)	-5.90 (p<.0001)	-5.97 (p<.0001)	A = 0.16 (p=.9368)
None-Current Diff	-11.97 (p<.0001)	-19.85 (p<.0001)	-15.91 (p<.0001)	B = -7.88 (p=.0095)
Future-Current Diff	-5.91 (p=.0226)	-13.95 (p<.0001)	-9.93 (p<.0001)	C = -8.04 (p=.0189)

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t
Cell means (simple means)					
Cognition for Men None	29.0701	0.7485	541	38.84	<.0001
Cognition for Women None	26.1946	0.6388	541	41.00	<.0001
Cognition for Men Future	23.0142	1.4928	541	15.42	<.0001
Cognition for Women Future	20.3029	1.1186	541	18.15	<.0001
Cognition for Men Current	17.0994	2.1402	541	7.99	<.0001
Cognition for Women Current	6.3487	1.9479	541	3.26	0.0012
Simple effects of sex					
Sex Difference for None	-2.8756	1.0112	541	-2.84	0.0046
Sex Difference for Future	-2.7113	1.8741	541	-1.45	0.1485
Sex Difference for Current	-10.7507	2.8993	541	-3.71	0.0002
Simple effects of dementia group					
None-Future Difference for Men	-6.0559	1.6351	541	-3.70	0.0002
None-Future Difference for Women	-5.8916	1.2778	541	-4.61	<.0001
None-Current Difference for Men	-11.9707	2.2450	541	-5.33	<.0001
None-Current Difference for Women	-19.8458	2.0286	541	-9.78	<.0001
Future-Current Difference for Men	-5.9148	2.5868	541	-2.29	0.0226
Future-Current Difference for Women	-13.9542	2.2389	541	-6.23	<.0001
Differences in simple effects = interactions					
A: Sex Effect differ between None and Future?	0.1643	2.0705	541	0.08	0.9368
A: None-Future Effect differ by Sex?	0.1643	2.0705	541	0.08	0.9368
B: Sex Effect differ between None and Current?	-7.8751	3.0245	541	-2.60	0.0095
B: None-Current Effect differ by Sex?	-7.8751	3.0245	541	-2.60	0.0095
C: Sex Effect differ between Future and Current?	-8.0394	3.4152	541	-2.35	0.0189
C: Future-Current Effect differ by Sex?	-8.0394	3.4152	541	-2.35	0.0189

Label	Contrasts		Chi-Square	F Value	Pr > ChiSq	Pr > F
	Num	Den				
Model R2 Test	8	541	230.14	28.77	<.0001	<.0001
Omnibus Dementia*Sex Interaction Test	2	541	6.98	3.49	0.0304	0.0311

SAS syntax and output: sex is continuous (dummy code) but dementia group is categorical:

```
TITLE2 'Continuous Sex (0=Men) and Categorical Dementia (Ref=Current)';
PROC MIXED DATA=work.Chapter2 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
CLASS demgroup; * Demgroup is categorical;
FORMAT demgroup Fdemgroup.; * Use my value label in the output;
MODEL cognition = age85 grip9 age85*grip9 sexMW demgroup
sexMW*demgroup / SOLUTION CHISQ DDFM=BW;
CONTRAST 'Model R2 Test' age85 1, grip9 1, age85*grip9 1, sexMW 1, demgroup -1 1 0,
demgroup -1 0 1, sexMW*demgroup -1 1 0, sexMW*demgroup -1 0 1 / CHISQ;
* Marginal dementia group mean differences averaged across sex means (what F-tests give);
LSMEANS demgroup / DIFF=ALL AT(age85 grip9 sexMW) = (0 0 .5);
* Simple effects of dementia group diffs per sex;
LSMEANS demgroup / DIFF=ALL AT(age85 grip9 sexMW) = (0 0 0); * For men;
LSMEANS demgroup / DIFF=ALL AT(age85 grip9 sexMW) = (0 0 1); * For women;
```

Solution for Fixed Effects

Effect	dementia group	Estimate	Standard Error	DF	t Value	Pr > t
Intercept		17.0994	2.1402	541	7.99	<.0001
age85		-0.3348	0.1199	541	-2.79	0.0054
grip9		0.6179	0.1481	541	4.17	<.0001
age85*grip9		0.1222	0.04035	541	3.03	0.0026
sexMW		-10.7507	2.8993	541	-3.71	0.0002
demgroup	1None	11.9707	2.2450	541	5.33	<.0001
demgroup	2Future	5.9148	2.5868	541	2.29	0.0226
demgroup	3Current	0
sexMW*demgroup	1None	7.8751	3.0245	541	2.60	0.0095
sexMW*demgroup	2Future	8.0394	3.4152	541	2.35	0.0189
sexMW*demgroup	3Current	0

SAS-created dummy code for ref group is omitted (= dots)

Type 3 Tests of Fixed Effects

Effect	Num	Den	Chi-Square	F Value	Pr > ChiSq	Pr > F
age85	1	541	7.80	7.80	0.0052	0.0054
grip9	1	541	17.41	17.41	<.0001	<.0001
age85*grip9	1	541	9.16	9.16	0.0025	0.0026
sexMW	1	541	19.45	19.45	<.0001	<.0001
demgroup	2	541	37.38	18.69	<.0001	<.0001
sexMW*demgroup	2	541	6.98	3.49	0.0304	0.0311

Demgroup effects are df=2 now
Main effect of sex is now **marginal** across dementia, but main effect of dementia is for men

Least Squares Means

Effect	dementia group	age85	grip9	sexMW	Estimate	Standard Error	DF	t Value	Pr > t
demgroup	1None	0.00	0.00	0.50	27.6323	0.4780	541	57.80	<.0001
demgroup	2Future	0.00	0.00	0.50	21.6586	0.9283	541	23.33	<.0001
demgroup	3Current	0.00	0.00	0.50	11.7241	1.4443	541	8.12	<.0001
demgroup	1None	0.00	0.00	0.00	29.0701	0.7485	541	38.84	<.0001
demgroup	2Future	0.00	0.00	0.00	23.0142	1.4928	541	15.42	<.0001
demgroup	3Current	0.00	0.00	0.00	17.0994	2.1402	541	7.99	<.0001
demgroup	1None	0.00	0.00	1.00	26.1946	0.6388	541	41.00	<.0001
demgroup	2Future	0.00	0.00	1.00	20.3029	1.1186	541	18.15	<.0001
demgroup	3Current	0.00	0.00	1.00	6.3487	1.9479	541	3.26	0.0012

Differences of Least Squares Means

Effect	dementia group		age85	grip9	sexMW	Estimate	Standard Error	DF	t Value	Pr > t
	NFC	NFC								
demgroup	1None	2Future	0.00	0.00	0.50	5.9738	1.0399	541	5.74	<.0001
demgroup	1None	3Current	0.00	0.00	0.50	15.9083	1.5134	541	10.51	<.0001
demgroup	2Future	3Current	0.00	0.00	0.50	9.9345	1.7135	541	5.80	<.0001
demgroup	1None	2Future	0.00	0.00	0.00	6.0559	1.6351	541	3.70	0.0002
demgroup	1None	3Current	0.00	0.00	0.00	11.9707	2.2450	541	5.33	<.0001
demgroup	2Future	3Current	0.00	0.00	0.00	5.9148	2.5868	541	2.29	0.0226
demgroup	1None	2Future	0.00	0.00	1.00	5.8916	1.2778	541	4.61	<.0001
demgroup	1None	3Current	0.00	0.00	1.00	19.8458	2.0286	541	9.78	<.0001
demgroup	2Future	3Current	0.00	0.00	1.00	13.9542	2.2389	541	6.23	<.0001

* Simple effects of sex per dementia group;

```
ESTIMATE 'Sex Diff for None'          demgroup 0 0 0 sexMW 1 sexMW*demgroup 1 0 0;
ESTIMATE 'Sex Diff for Future'        demgroup 0 0 0 sexMW 1 sexMW*demgroup 0 1 0;
ESTIMATE 'Sex Diff for Current'       demgroup 0 0 0 sexMW 1 sexMW*demgroup 0 0 1;
```

* Simple effects of dementia group per sex (redundant with LSMEANS);

```
ESTIMATE 'None-Future Diff for Men'   demgroup -1 1 0 sexMW 0 sexMW*demgroup 0 0 0;
ESTIMATE 'None-Future Diff for Women' demgroup -1 1 0 sexMW 0 sexMW*demgroup -1 1 0;
ESTIMATE 'None-Current Diff for Men'  demgroup -1 0 1 sexMW 0 sexMW*demgroup 0 0 0;
ESTIMATE 'None-Current Diff for Women' demgroup -1 0 1 sexMW 0 sexMW*demgroup -1 0 1;
ESTIMATE 'Future-Current Diff for Men' demgroup 0 -1 1 sexMW 0 sexMW*demgroup 0 0 0;
ESTIMATE 'Future-Current Diff for Women' demgroup 0 -1 1 sexMW 0 sexMW*demgroup 0 -1 1;
```

* Differences in simple effects = interactions;

```
ESTIMATE 'A: Sex Effect differ between None and Future?' sexMW*demgroup -1 1 0;
ESTIMATE 'A: None-Future Effect differ by Sex?'          sexMW*demgroup -1 1 0;
ESTIMATE 'B: Sex Effect differ between None and Current?' sexMW*demgroup -1 0 1;
ESTIMATE 'B: None-Current Effect differ by Sex?'         sexMW*demgroup -1 0 1;
ESTIMATE 'C: Sex Effect differ between Future and Current?' sexMW*demgroup 0 -1 1;
ESTIMATE 'C: Future-Current Effect differ by Sex?'       sexMW*demgroup 0 -1 1;
```

RUN; TITLE2;

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t
Sex Difference for None	-2.8756	1.0112	541	-2.84	0.0046
Sex Difference for Future	-2.7113	1.8741	541	-1.45	0.1485
Sex Difference for Current	-10.7507	2.8993	541	-3.71	0.0002
None-Future Difference for Men	-6.0559	1.6351	541	-3.70	0.0002
None-Future Difference for Women	-5.8916	1.2778	541	-4.61	<.0001
None-Current Difference for Men	-11.9707	2.2450	541	-5.33	<.0001
None-Current Difference for Women	-19.8458	2.0286	541	-9.78	<.0001
Future-Current Difference for Men	-5.9148	2.5868	541	-2.29	0.0226
Future-Current Difference for Women	-13.9542	2.2389	541	-6.23	<.0001
A: Sex Effect differ between None and Future?	0.1643	2.0705	541	0.08	0.9368
A: None-Future Effect differ by Sex?	0.1643	2.0705	541	0.08	0.9368
B: Sex Effect differ between None and Current?	-7.8751	3.0245	541	-2.60	0.0095
B: None-Current Effect differ by Sex?	-7.8751	3.0245	541	-2.60	0.0095
C: Sex Effect differ between Future and Current?	-8.0394	3.4152	541	-2.35	0.0189
C: Future-Current Effect differ by Sex?	-8.0394	3.4152	541	-2.35	0.0189

SAS syntax and default output: sex is categorical and dementia group is categorical:

```
TITLE2 'Categorical Sex (Ref=Women) and Categorical Dementia (Ref=Current)';
PROC MIXED DATA=work.Chapter2 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
CLASS sexMW demgroup;
FORMAT sexMW Fsex. demgroup Fdemgroup.; * Use my value labels in the output;
MODEL cognition = age85 grip9 age85*grip9 sexMW demgroup
sexMW*demgroup / SOLUTION CHISQ DDFM=BW;
CONTRAST 'Model R2 Test' age85 1, grip9 1, age85*grip9 1, sexMW -1 1, demgroup -1 1 0,
demgroup -1 0 1, sexMW*demgroup -1 1 0 1 -1 0, sexMW*demgroup -1 0 1 1 0 -1 /CHISQ;
* Marginal sex means and tests diffs averaged over demgroup (F-test below);
LSMEANS sexMW / DIFF=ALL AT(age85 grip9) = (0 0);
* Marginal demgroup means and tests of diffs averaged over sex (F-test below);
LSMEANS demgroup / DIFF=ALL AT(age85 grip9) = (0 0);
```

Solution for Fixed Effects

Effect	sexMW: Sex (0=M, 1=W)	dementia group NFC	Estimate	Standard Error	DF	t Value	Pr > t
Intercept			6.3487	1.9479	541	3.26	0.0012
age85			-0.3348	0.1199	541	-2.79	0.0054
grip9			0.6179	0.1481	541	4.17	<.0001
age85*grip9			0.1222	0.04035	541	3.03	0.0026
sexMW	0Men		10.7507	2.8993	541	3.71	0.0002
sexMW	1Women		0
demgroup		1None	19.8458	2.0286	541	9.78	<.0001
demgroup		2Future	13.9542	2.2389	541	6.23	<.0001
demgroup		3Current	0
sexMW*demgroup	0Men	1None	-7.8751	3.0245	541	-2.60	0.0095
sexMW*demgroup	0Men	2Future	-8.0394	3.4152	541	-2.35	0.0189
sexMW*demgroup	0Men	3Current	0
sexMW*demgroup	1Women	1None	0
sexMW*demgroup	1Women	2Future	0
sexMW*demgroup	1Women	3Current	0

SAS-created dummy code for ref groups are omitted (= dots)

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
age85	1	541	7.80	7.80	0.0052	0.0054
grip9	1	541	17.41	17.41	<.0001	<.0001
age85*grip9	1	541	9.16	9.16	0.0025	0.0026
sexMW	1	541	19.45	19.45	<.0001	<.0001
demgroup	2	541	129.24	64.62	<.0001	<.0001
sexMW*demgroup	2	541	6.98	3.49	0.0304	0.0311

Both main effects of sex and demgroup are now **marginal** across the other

Least Squares Means

Effect	sexMW (0=M, 1=W)	dementia group NFC	age85	grip9	Estimate	Standard Error	DF	t Value	Pr > t
sexMW	0Men		0.00	0.00	23.0613	0.9251	541	24.93	<.0001
sexMW	1Women		0.00	0.00	17.6154	0.7888	541	22.33	<.0001
demgroup		1None	0.00	0.00	27.6323	0.4780	541	57.80	<.0001
demgroup		2Future	0.00	0.00	21.6586	0.9283	541	23.33	<.0001
demgroup		3Current	0.00	0.00	11.7241	1.4443	541	8.12	<.0001

Differences of Least Squares Means (Marginal effects)										
Effect	SexMW (0=M, 1=W)	dementia group	SexMW (0=M, 1=W)	dementia group	age85	grip9	Estimate	Standard Error	Value	Pr > t
sexMW	0Men		1Women		0.00	0.00	5.4459	1.2349	4.41	<.0001
demgroup		1None		2Future	0.00	0.00	5.9738	1.0399	5.74	<.0001
demgroup		1None		3Current	0.00	0.00	15.9083	1.5134	10.51	<.0001
demgroup		2Future		3Current	0.00	0.00	9.9345	1.7135	5.80	<.0001

Dementia Group Means	Men Mean	Women Mean	MARGINAL MEAN	Sex Difference
None Mean	29.07	26.20	27.62	-2.87 (p=.0046)
Future Mean	23.01	20.30	21.66	-2.71 (p=.1485)
Current Mean	<u>17.10</u>	<u>6.35</u>	11.72	-10.75 (p=.0002)
MARGINAL	23.03	17.62		-5.45 (p<.0001)

Dementia Group Differences	Men Diff	Women Diff	MARGINAL DIFF	Simple Effect Difference
None-Future Diff	-6.06 (p=.0002)	-5.90 (p<.0001)	-5.97 (p<.0001)	A=0.16 (p=.9368)
None-Current Diff	-11.97 (p<.0001)	-19.85 (p<.0001)	-15.91 (p<.0001)	B= -7.88 (p=.0095)
Future-Current Diff	-5.91 (p=.0226)	-13.95 (p<.0001)	-9.93 (p<.0001)	C= -8.04 (p=.0189)

So what we know from the default (ANOVA) output (bolded in the table below) is that the effect of sex differs somehow by dementia group (but not whether there is an effect of sex within each dementia group) OR that the effect of dementia differs somehow by sex (but not whether there is an effect of dementia per sex). Super helpful, huh? This is why we need the rest of our code. To find out the differences within condition (simple effects), some ANOVA-trained-only people may turn to separate models (e.g., select only men, then examine the effect of dementia group). However, that model will have a different residual variance, and thus the test may not be the same as it would be when done correctly in a full model. So let's see how to ask for all the relevant simple effects instead...

```
* Cell means and simple effect tests (SLICE means per);
LSMEANS sexMW*demgroup / SLICE=demgroup SLICE=sexMW DIFF=ALL AT(age85 grip9) = (0 0);
```

Least Squares Means (cell means)										
Effect	SexMW (0=M, 1=W)	dementia group	age85	grip9	Estimate	Standard Error	DF	t Value	Pr > t	
sexMW*demgroup	0Men	1None	0.00	0.00	29.0701	0.7485	541	38.84	<.0001	
sexMW*demgroup	0Men	2Future	0.00	0.00	23.0142	1.4928	541	15.42	<.0001	
sexMW*demgroup	0Men	3Current	0.00	0.00	17.0994	2.1402	541	7.99	<.0001	
sexMW*demgroup	1Women	1None	0.00	0.00	26.1946	0.6388	541	41.00	<.0001	
sexMW*demgroup	1Women	2Future	0.00	0.00	20.3029	1.1186	541	18.15	<.0001	
sexMW*demgroup	1Women	3Current	0.00	0.00	6.3487	1.9479	541	3.26	0.0012	

Differences of Least Squares Means (every possible cell mean difference)										
Effect	SexMW (0=M, 1=W)	dementia group	SexMW (0=M, 1=W)	dementia group	age85	grip9	Estimate	Standard Error	Value	Pr > t
sexMW*demgroup	0Men	1None	0Men	2Future	0.00	0.00	6.0559	1.6351	3.70	0.0002
sexMW*demgroup	0Men	1None	0Men	3Current	0.00	0.00	11.9707	2.2450	5.33	<.0001
sexMW*demgroup	0Men	1None	1Women	1None	0.00	0.00	2.8756	1.0112	2.84	0.0046
sexMW*demgroup	0Men	1None	1Women	2Future	0.00	0.00	8.7672	1.3522	6.48	<.0001
sexMW*demgroup	0Men	1None	1Women	3Current	0.00	0.00	22.7214	2.0981	10.83	<.0001

sexMW*demgroup	OMen	2Future	OMen	3Current	0.00	0.00	5.9148	2.5868	2.29	0.0226
sexMW*demgroup	OMen	2Future	1Women	1None	0.00	0.00	-3.1803	1.6509	-1.93	0.0546
sexMW*demgroup	OMen	2Future	1Women	2Future	0.00	0.00	2.7113	1.8741	1.45	0.1485
sexMW*demgroup	OMen	2Future	1Women	3Current	0.00	0.00	16.6655	2.4724	6.74	<.0001
sexMW*demgroup	OMen	3Current	1Women	1None	0.00	0.00	-9.0951	2.2436	-4.05	<.0001
sexMW*demgroup	OMen	3Current	1Women	2Future	0.00	0.00	-3.2035	2.4179	-1.32	0.1858
sexMW*demgroup	OMen	3Current	1Women	3Current	0.00	0.00	10.7507	2.8993	3.71	0.0002
sexMW*demgroup	1Women	1None	1Women	2Future	0.00	0.00	5.8916	1.2778	4.61	<.0001
sexMW*demgroup	1Women	1None	1Women	3Current	0.00	0.00	19.8458	2.0286	9.78	<.0001
sexMW*demgroup	1Women	2Future	1Women	3Current	0.00	0.00	13.9542	2.2389	6.23	<.0001

Tests of Effect Slices (simple effects)

Effect	SexMW (0=M, 1=W)	dementia group			Num DF	Den DF	F Value	Pr > F
		NFC	age85	grip9				
sexMW*demgroup		1None	0.00	0.00	1	541	8.09	0.0046
sexMW*demgroup		2Future	0.00	0.00	1	541	2.09	0.1485
sexMW*demgroup		3Current	0.00	0.00	1	541	13.75	0.0002
sexMW*demgroup	OMen		0.00	0.00	2	541	18.69	<.0001
sexMW*demgroup	1Women		0.00	0.00	2	541	53.16	<.0001

* Order by CLASS statement: MN MF MC WN WF WC -- check order with / E;

* Simple effects of sex per dementia group (redundant with LSMEANS);

ESTIMATE 'Sex Diff for None' demgroup 0 0 0 sexMW -1 1 sexMW*demgroup -1 0 0 1 0 0;

ESTIMATE 'Sex Diff for Future' demgroup 0 0 0 sexMW -1 1 sexMW*demgroup 0 -1 0 0 1 0;

ESTIMATE 'Sex Diff for Current' demgroup 0 0 0 sexMW -1 1 sexMW*demgroup 0 0 -1 0 0 1;

* Simple effects of dementia group per sex (redundant with LSMEANS);

ESTIMATE 'None-Future Diff for Men' demgroup -1 1 0 sexMW 0 0 sexMW*demgroup -1 1 0 0 0 0;

ESTIMATE 'None-Future Diff for Women' demgroup -1 1 0 sexMW 0 0 sexMW*demgroup 0 0 0 -1 1 0;

ESTIMATE 'None-Current Diff for Men' demgroup -1 0 1 sexMW 0 0 sexMW*demgroup -1 0 1 0 0 0;

ESTIMATE 'None-Current Diff for Women' demgroup -1 0 1 sexMW 0 0 sexMW*demgroup 0 0 0 -1 0 1;

ESTIMATE 'Future-Current Diff for Men' demgroup 0 -1 1 sexMW 0 0 sexMW*demgroup 0 -1 1 0 0 0;

ESTIMATE 'Future-Current Diff for Women' demgroup 0 -1 1 sexMW 0 0 sexMW*demgroup 0 0 0 0 -1 1;

* Simple effect Diffs = interactions;

ESTIMATE "A: Sex Effect differ between None and Future?" sexMW*demgroup -1 1 0 1 -1 0;

ESTIMATE "A: None-Future Effect differ by Sex?" sexMW*demgroup -1 1 0 1 -1 0;

ESTIMATE "B: Sex Effect differ between None and Current?" sexMW*demgroup -1 0 1 1 0 -1;

ESTIMATE "B: None-Current Effect differ by Sex?" sexMW*demgroup -1 0 1 1 0 -1;

ESTIMATE "C: Sex Effect differ between Future and Current?" sexMW*demgroup 0 -1 1 0 1 -1;

ESTIMATE "C: Future-Current Effect differ by Sex?" sexMW*demgroup 0 -1 1 0 1 -1;

RUN; TITLE2;

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t
Sex Difference for None	-2.8756	1.0112	541	-2.84	0.0046
Sex Difference for Future	-2.7113	1.8741	541	-1.45	0.1485
Sex Difference for Current	-10.7507	2.8993	541	-3.71	0.0002
None-Future Difference for Men	-6.0559	1.6351	541	-3.70	0.0002
None-Future Difference for Women	-5.8916	1.2778	541	-4.61	<.0001
None-Current Difference for Men	-11.9707	2.2450	541	-5.33	<.0001
None-Current Difference for Women	-19.8458	2.0286	541	-9.78	<.0001
Future-Current Difference for Men	-5.9148	2.5868	541	-2.29	0.0226
Future-Current Difference for Women	-13.9542	2.2389	541	-6.23	<.0001
A: Sex Effect differ between None and Future?	-0.1643	2.0705	541	-0.08	0.9368
A: None-Future Effect differ by Sex?	-0.1643	2.0705	541	-0.08	0.9368
B: Sex Effect differ between None and Current?	7.8751	3.0245	541	2.60	0.0095
B: None-Current Effect differ by Sex?	7.8751	3.0245	541	2.60	0.0095
C: Sex Effect differ between Future and Current?	8.0394	3.4152	541	2.35	0.0189
C: Future-Current Effect differ by Sex?	8.0394	3.4152	541	2.35	0.0189

Moral of the story: Given a significant interaction, it is very likely that the marginal tests provided by default will be misleading and/or will not help you test any meaningful hypotheses. Get simple effects!